

# Formula (6.16) Revisited: Obtaining the Generating Function by a Holonomic Differential Equation (or: no "tricks" anymore...)

Folkmar Bornemann, March 9, 2007

We apply Zeilberger's algorithm to the return probabilities  $p_k$  to obtain a holonomic recurrence equation. (Writing  $a = p_{EW}$  and  $b = p_{NS}$  for short.) This is, once more, what we have done in the Maple session preceeding (6.14).

```
> rec:=sumtools[sumrecursion](binomial(2*k,k)*binomial(k,j)^2*a^(2*j)*b^(2*(k-j)),j,p(k));
```

$$rec := 4(a-b)^2(a+b)^2(2k-1)(2k-3)p(k-2) - 2(a^2+b^2)(2k-1)^2p(k-1) + p(k)k^2$$

Now, we depart and transform the holonomic recurrence for  $p_k$  into a holonomic differential

equation for the generating function  $E(x) = \sum_{k=0}^{\infty} p_k x^k$ .

```
> dic:=gfun[rectodiffeq]({rec=0,p(0)=1,p(1)=2*(a^2+b^2)},p(k),E(x))
```

```
: deq:=remove(hastype,dic,`= `); ic:=select(hastype,dic,`= `);
```

$$\begin{aligned} \text{deq} := & \{ (12 x a^4 - 24 x b^2 a^2 + 12 x b^4 - 2 a^2 - 2 b^2) E(x) \\ & + (-96 x^2 b^2 a^2 + 48 x^2 a^4 + 48 x^2 b^4 - 16 x a^2 - 16 x b^2 + 1) \left( \frac{d}{dx} E(x) \right) \\ & + (16 x^3 b^4 - 32 x^3 b^2 a^2 + 16 x^3 a^4 - 8 x^2 b^2 - 8 x^2 a^2 + x) \left( \frac{d^2}{dx^2} E(x) \right) \} \\ \text{ic} := & \{ D(E)(0) = 2 a^2 + 2 b^2, E(0) = 1 \} \end{aligned}$$

The differential equation degenerates for  $x = 0$ , but the initial value  $\frac{d}{dx} E(0) = 2(a^2 + b^2)$  is by construction compatible with this degeneracies.

```
> subs(x=0,deq);
```

$$\{ (-2 a^2 - 2 b^2) E(0) + \text{diff}(E(0), 0) \}$$

The second order holonomic differential equation is easily solved *numerically* for the desired expected value  $E$  of the number of visits by integrating it to  $x = 1$ . This yields a further numerical method to solve Problem 6 in less than a second (see the accompanying Matlab routine 'problem6\_by\_diffeq.m'). However, we could courageously try to find the solution to it using *symbolic* methods.

```
> sol:=DEtools[hypergeomsols](deq,E(x));
```

`sol :=`

$$\left[ \frac{\text{LegendreP}\left(\frac{-1}{2}, \frac{24 x b a + 4 x a^2 + 4 x b^2 - 1}{4 x a^2 - 8 x b a + 4 x b^2 - 1}\right)}{\sqrt{4 x a^2 - 8 x b a + 4 x b^2 - 1}}, \frac{\text{LegendreQ}\left(\frac{-1}{2}, \frac{24 x b a + 4 x a^2 + 4 x b^2 - 1}{4 x a^2 - 8 x b a + 4 x b^2 - 1}\right)}{\sqrt{4 x a^2 - 8 x b a + 4 x b^2 - 1}} \right]$$

Because of the degeneracy, there is a one-dimensional subspace of solutions that have finite values at  $x = 0$ . Lets check the given basis elements for that property.

`> eval(subs(x=0, sol[1]));`

$-I$

`> eval(subs(x=0, sol[2]));`

Error, (in LegendreQ) numeric exception: division by zero

In half of the runs (Maple behaves non-deterministic here), we are lucky and the first element of the given basis, involving the *LegendreP* function, is finite at  $x = 0$ . (Otherwise restart and repeat.) We normalize for  $E(0) = 1$ .

`> sol[1]/eval(subs(x=0, sol[1]));`

$$\frac{\text{LegendreP}\left(\frac{-1}{2}, \frac{24 x b a + 4 x a^2 + 4 x b^2 - 1}{4 x a^2 - 8 x b a + 4 x b^2 - 1}\right) I}{\sqrt{4 x a^2 - 8 x b a + 4 x b^2 - 1}}$$

[ Cross-check that the differential equation is satisfied.

[ > **simplify(subs(E(x)=%,deq));**

{0}

[ The expression that we have obtained already qualifies as a "closed" form solution. However, we can even do better by recognizing that LegendreP $\left(-\frac{1}{2}, z\right)$  is related to the complete elliptic integral of the first kind  $K$  (see Abramowitz/Stegun §8.13). The desired expected value  $E$  of the number of visits is obtained by evaluating at  $x = 1$ .

[ > **E=simplify(subs(x=1,convert(%%,EllipticK))) assuming  
0<=b,b<=a,a-b<=1/2;**

$$E = \frac{2 \operatorname{EllipticK}\left(\frac{4\sqrt{b}\sqrt{a}}{\sqrt{1-4a^2+8ba-4b^2}}\right)}{\sqrt{1-4a^2+8ba-4b^2} \pi}$$

[ >

[ This is exactly formula (6.16) in §6.5, that we had previously obtained by a "trick" due to Herb Wilf. Our approach here is probably more systematic.